GTAP 데이터베이스를 활용한 GAMS-MPSGE 모형 작성

이창수 (경희대), 송백훈 (성신여대)

2012. 08. 24.

1. 연구의 개요

▣ 연구의 필요성 및 목적

□ 플랫폼 형태의 유연한 KIEP 표준모형 구축의 필요성

○ 각종 FTA 및 DDA 협상의 진전과 환경, ODA 등 새로운 이슈의 부상에 따라 정부가 specific한 주제에 대한 경제적 효과분석을 수시로 요구하고 있으나, 수정 용이한 KIEP 표준모형이 구축되어 있지 않아, 각종 효과에 대한 분석이 플랫폼 모형을 기반으로 일관성 있게 그리고 상호 비교가능하게 이루어지고 있지 못함.

○ KIEP는 GTAP-GEMPACK을 사용하는 CGE 모형 (static 및 recursive dynamic model)을 분석하여 정부의 요구에 부응하고 있으나, (1) GTAP 표준모형의 수정이 유연하지 않을 뿐 아니라 (2) 표준모형의 분석 coverage가 경직적이어서, 플랫폼 표준모형의 변형을 통한 specific한 쟁점 분석이 어려움. (3) 또한 완전동태 분석 (intertemporal dynamic model)이 불가능함.

○ 즉, GTAP-GEMPACK은 (1) 프로그램을 자료관리, 마감조건 변경, 정책시뮬레이션 시나리오 작성 및 결과보고 등으로 분절하고 있어, 사용자의 source code 접근 및 수정이 매우 경직적임. 이에 따라 다양한 주제 및 쟁점에 대한 유연하고 일관성 있는 분석이 사실상 어려운 상황임.

□ GTAP 데이터베이스를 사용하여 Global SAM (Social Accounting Matrix)을 만들고, GAMS-MPSGE 프로그램를 활용하여 플랫폼 형태의 KIEP 표준모형을 coding하고자 함.

○ GTAP 표준모형과 equivalent한 GAMS-MPSGE 모형 작성

○ GAMS-MPSGE 특성 상 플랫폼 모형의 수정으로, 다양한 주제에 적합한 변종모형 구축과 쟁점분석이 가능하게 됨. 이에 따라 다양한 정책 수요에 신속하고 유연하게 대응하게 될 것임.

▣ GTAP-GEMPACK CGE 분석과 GAPT-GAMS CGE 분석의 차별성

구 분	CGE 분석의 차별성	
	장점	단점
GTAP-GEMPACK	-호주 ORANI 모형을 기반으로 구축된 GTAP 기본모형과 함께 GTAP (국제산업연관표) 데이터베이스를 제공함으로써, DB 구축이 쉽고 모형 접근이 용이함 -많은 방정식과 변수를 가진 대규모 모형으로 다국 대규모 정책실험 가능	-다양한 시나리오에 대한 유연한 분석과 신속하고 눈에 보이는 결과보고를 사실상 차단 o 프로그램을 자료관리, 마감조건 변경, 정책시뮬레이션 시나리오 작성, 결과보고 등으로 분절하여 사용자의 프로그램접근을 부분적으로만 허용 o 결과를 출력함에 있어서도 정해진 양식에 따라 규정된 것만을 보여주고 있음, -많은 방정식과 변수를 가진 대규모 모형으로 모형수정이 어려워 독자모형을 구축하기가 어려움. o 전세계적으로 표준모형 source file을 자유롭게 변경할 수 있는 최고 수준의 연구자가 별로 없음.
GAMS	-GAMS CGE 모형의 경우 모든 연립방정식 체계를 직접 입력하여 작업하기 때문에 프로그램 그 자체가 독자 모형이며, 사용자의 필요에 의해 쉽게 수정할 수 있는 유연성의 강점이 있음	-GTAP-GEMPACK과 달리 SAM(social accounting matrix)을 직접 구축해야 함. o 1국 또는 2국모형의 경우 산업연관표를 활용하여 SAM을 구축할 수 있지만 다국모형의 경우 국제산업연관표 작성 자체가 어려워 GAMS를 활용한 정책실험이 곤란함.
GTAP-GAMS	- GTAP-GEMPACK CGE의 DB장점과 GAMS CGE의 유연성 및 신속성 장점을 결합한 시도임. o GTAP 데이터베이스(의 편리성)를 사용하여 GAMS에서 신축적으로 프로그램하는 것으로 독자모형 구축이 용이함. o 자료처리, 균형식체계 구축, 정책실험 및 결과출력을 한 source 프로그램에서 일괄처리 o 표준화된 KIEP 모형 구축후 제3자가 쉽게 활용할 수 있음. o Calibrartion 과정의 투명화: CD, CES 등의 계수 및 탄력성 값을 SAM 자료를 사용하여 직접 계산	-GEMPACK 모형에 대응하는 GTAP 데이터의 코드(HAR)를 GAMS 모형에 적합하게 수정하여 GAMS에서 읽을 수 있게(GDX) 전환하는 module을 구축해야 함. -GAMS 프로그램 접근이 어려움
GTAP-GAMS-MPSGE (본연구)	-기본적으로 GTAP-GAMS와 동일 -modeling part만 상이. 모든 연립방정식 체계를 직접 입력할 필요가 없음. MPSGE framework에 맞추어 tree 구조를 결정하고 대체탄력성만 지정하면 modeling 완료 -inter-temporal dynamic 분석이 쉬움.	-기본적으로 GTAP-GAMS와 동일 -MPSGE modeling에 대한 이해 필요

▣ 연구 방법

□ global SAM: 통계분석 및 전환

○ GTAP DB를 활용하여 global SAM 작성

□ MPSGE code로 KIEP CGE 모형 작성

○ CES, CET, LES 등 각종 함수의 계수 및 탄력성을 기준연도 SAM DB와 GTAP 탄력성을 사용하여 calibration and replication.
○ 플랫폼 표준모형 (static, recursive dynamic, full dynamic): 과거 GTAP 표준모형이 분석했던 FTA, DDA 효과 등 표준적인 정책실험을 수행
○ ODA, 환경 등 specific한 수요를 반영한 변종모형 (static, recursive dynamic, full dynamic)의 작성 및 정책실험
○ 다양한 분석 시나리오를 일괄작업으로 신속히 simulation.

▣ 주요 연구내용

□ GTAP DB를 활용한 Global SAM 작성

□ 플랫폼 KIEP 모형 작성
○ static model
○ recursive dynamic model
○ inter-temporal dynamic model

□ 정책실험
○ 한-중, 한-중-일 FTA의 경제적 효과

▣ 기대 효과

□ GTAP 표준모형과 equivalent한 KIEP 플랫폼 CGE 분석틀을 구축

2. DGE (Dynamic General Equilibrium) and GAMS-MPSGE

가. 램지모형의 동태적 최적화 문제

constant-elasticity-of-intertemporal-substitution (CEIS) utility function

$$\max \quad \sum_{t=0}^{\infty}\left(\frac{1}{1+\rho}\right)^{t}\frac{C_t^{1-\theta}-1}{1-\theta}$$

$$\text{s.t.} \quad \begin{aligned} C_t &= f(K_t) - I_t \\ K_{t+1} &= (1-\delta)K_t + I_t \\ K_0 &= \overline{K_0} \end{aligned}$$

MPSGE 램지모형에서 총공급은 자본스톡의 함수로 표현 가능.

$$Y_t = F(\overline{L_t}, K_t) = f(K_t)$$

생산함수가 위와 같이 정의될 때, 단위비용함수는 다음과 같음.

$$c\left(p_t^L, r_t^K\right) \equiv \min p_t^L a_L + \min r_t^K a_K$$

$$\text{s.t.} \quad F(a_L, a_K) = 1$$

쉐퍼드 렘마에 의하면 노동과 자본의 보상수요는 단위비용함수의 편미분 값임.

$$a_K\left(r^K, p^L\right) = \frac{\partial c\left(p_t^L, r_t^K\right)}{\partial r_t^K},$$

$$a_L\left(r^K, p^L\right) = \frac{\partial c\left(p_t^L, r_t^K\right)}{\partial p_t^L}.$$

나. GAMS/MPSGE 모형에서의 일반균형 연립방정식 체계

다음과 같이 3개 군으로 분류

(1) 시장청산조건과 관련 시장가격

- 산출물 시장 (시장가격 p_t):

$$Y_t = C_t(p, M) + I_t$$

- 노동시장 (시장가격 p_t^L):

$$\overline{L_t} = a_L(r_t^K, p_t^L)\, Y_t$$

- 자본서비스시장 (시장가격 r_t^K):

$$K_t = a_K(r_t^K, p_t^L)\, Y_t$$

- 자본스톡 (자본구입가격 p_t^K):

$$K_{t+1} = (1-\delta)K_t + I_t$$

(2) 영이윤 조건과 관련 수량변수

- 산출물 (Y_t):

$$p_t = c(r_t^K, p_t^L)$$

- 투자 ($I_t \geq 0$):

$$p_t \geq p_{t+1}^K$$

- 자본스톡 (K_t):

$$p_t^K = r_t^K + (1-\delta)\, p_{t+1}^K$$

(3) 소득정의 (Income balance)

$$M = p_0^K \overline{K_0} + \sum_{t=0}^{\infty} p_t^L \overline{L_t}$$

다. CEIS 선호체계

수요함수 $C_t(p, M)$의 구체화 필요

(i) Additively separable utility:

$$U(c) = \sum_{t=0}^{\infty} \left(\frac{1}{1+\rho} \right)^t \frac{C_t^{1-\theta} - 1}{1-\theta}$$

(ii) Linearly homogeneous utility:

$$\widehat{U(c)} = \left[\sum_{t=0}^{\infty} \left(\frac{1}{1+\rho} \right)^t C_t^{1-\theta} \right]^{\frac{1}{1-\theta}}$$

$\hat{U}$ 가 U의 단조증가 함수이기 때문에 $\hat{U}$와 U 는 동일한 수요함수를 도출함.
동일한 선호체계를 갖는 두 효용함수에서 한계대체율은 다음과 같음.

$$\frac{\partial U / \partial C_{t+1}}{\partial U / \partial C_t} = \frac{1}{1+\rho} \left(\frac{C_t}{C_{t+1}} \right)^{1-\theta}$$

(1) $\beta = \dfrac{1}{1+\rho}$ (MPSGE), $\dfrac{1}{\beta} - 1 = (1+\rho) - 1 = \rho$

즉 β: a discount factor (시간할인율)

ρ: the pure rate of time preference (순 시간선호율)

(2) $U(c) = \sum_{t=0}^{\infty} \beta^t \dfrac{c^{1-\theta}}{1-\theta}$ and $U(c) = \left[\sum_{t=0}^{\infty} \beta^t c^{1-\theta} \right]^{\frac{1}{1-\theta}}$

have same preferences (demand function)

이 경우 CES 효용함수의 대체탄력성 = 시점간 대체탄력성 (inter-temporal elasticity of substitution) $\sigma = \dfrac{1}{1-(1-\theta)} = \dfrac{1}{\theta}$

따라서 $\hat{U}$의 1% 후생변동은 소득 1% 변화와 동등함.

라. 무한기간 방정식체계의 유한기간 체계로의 근사

무한대 기간 방정식체계를 T 기간의 유한기간 체계로 근사 (a finite approximation of the model)시키기 위해, 소비자의 문제를 두 부분으로 구분.

무한시간 램지모델 (the infinite-horizon problem of the representative agent in Ramsey's model):

$$\begin{aligned} \max \quad & \sum_{t=0}^{\infty}\left(\frac{1}{1+\rho}\right)^{t} u(c_t) \\ \text{s.t.} \quad & \sum_{t=0}^{\infty} p_c c_t = p_0^K \overline{K_0} + \sum_{t=0}^{\infty} p_t^L \overline{L_t} \end{aligned}$$

최종기 자산가치를 다음과 같이 정의:

$$A_T^* = \sum_{t=T+1}^{\infty}\left(p_c c_t^* - p_t^L \overline{L_t}\right)$$

두 기간으로 구분될 때 대표 소비자의 효용극대화 문제:

$$\begin{aligned} \max \quad & \sum_{t=0}^{T}\left(\frac{1}{1+\rho}\right)^{t} u(c_t) + \sum_{t=T+1}^{\infty}\left(\frac{1}{1+\rho}\right)^{t} u(c_t) \\ \text{s.t.} \quad & \sum_{t=0}^{T} p_c c_t = p_0^K \overline{K_0} + \sum_{t=0}^{T} p_t^L \overline{L_t} - A_T \\ & \sum_{t=T+1}^{\infty} p_c c_t = A_T + \sum_{t=T+1}^{\infty} p_t^L \overline{L_t} \end{aligned}$$

A_T가 고정값을 갖는다면, 위 문제는 별도의 두 개 (T 기간 까지와 그 이후)의 최적화문제로 정의할 수 있음. 무한대 기간의 최적 해로 A_T^*를 정의할 수 있다면, 유한기간 모형에서의 0부터 T 기간까지의 소비수준은 무한기간 모형에서의 최적 소비수준과 동일하게 될 것임.

균형상태에서의 통제변수 I_T와 상태변수 K_{T+1}

- 최종기 투자증가율 = 장기균제상태성장률:

$$\frac{I_T}{I_{T-1}} = 1 + g$$

- 최종기 투자증가율 = 총산출 증가율:

$$\frac{I_T}{I_{T-1}} = \frac{Y_T}{Y_{T-1}}$$

- 최종기 투자증가율 = 소비증가율:

$$\frac{I_T}{I_{T-1}} = \frac{C_T}{C_{T-1}}$$

보정 (calibration): 이자율이 $\bar{r}$이고 성장률이 $\bar{g}$인 균제상태

(1) I_t의 영이윤 조건 ($p_t \geq p_{t+1}^K$)으로부터, 자본스톡 가격은

$$p_{t+1}^K = \frac{p_t^K}{(1+\bar{r})} = p_t$$

따라서 $$p_t^K = (1+\bar{r})p_t$$

따라서 기준연도 자본스톡 가격은

$$\overline{p^K} = (1+\bar{r})$$

(2) K_t 의 영이윤 조건에서 자본 서비스 가격 (r_t^K)을 결정.

$$p_t^K = r_t^K + (1-\delta)p_{t+1}^K$$

p_t^K 와 p_{t+1}^K를 대입하면 기준연도 자본 렌탈가격은 이자율과 감가상각율의 합.

$$r_t^K = \bar{r} + \delta$$

(3) 기준연도 자본소득, 투자, 장기균형 이자율과 감가상각률이 일치성을 갖도록 조정.
다음과 같은 1기 자본스톡의 시장청산 조건은

$$K_1 = (1+g)\overline{K_0} = (1-\delta)\overline{K_0} + \bar{I}$$

기준연도 자본스톡의 감가상각과 성장률에서 기준연도 투자가 계산될 수 있음을 시사.

$$\bar{I} = (g+\delta)\overline{K_0}$$

(4) $\overline{r^K}$ 를 사용하여 기준연도 자본소득 $\overline{VK}$를 기반으로 하여 자본스톡 $\overline{K_0}$ 를 결정.

$$\bar{I} = (g+\delta)\overline{K_0} = (g+\delta)\frac{\overline{VK}}{r+\delta} = \frac{g+\delta}{r+\delta}\overline{VK}$$

$$\bar{I} = \frac{g+\delta}{r+\delta}\overline{VK}$$

마. GAMS Code: Describing a Steady-State Equilibrium

```
$TITLE Ramsey Model -- MPSGE formulation calibrated to base year data

$ontext
Calibrate to the steady-state condition:

I0 = KD0 * (g + delta) / (r + delta)

where g=2, delta=7, r=5, so

I0 = 48 * 9 / 12 = 36

Y I FD
P 100 -36 -64
PL -52 52
RK -48 48
PS 36 -36

$offtext

PARAMETER g Growth rate /0.02/,
r Interest rate /0.05/,
delta Depreciation rate /0.07/,
kvs Capital value share /0.48/,
y0 Base year output,
kd0 Base year rental value of capital,
k0 Base year capital stock,
i0 Base year investment,
c0 Base year consumption,
l0 Base year labor input,
kstock Base year capital stock multiplier /1/,
```

```
taxk(t) Capital tax rate in period T,
qref(t) Reference quantity path,
pref(tt) Reference price path;

* Use the GAMS ORD (ordinality) and CARD (cardinality)
* functions to automate the identification of the first
* and last periods of the model horizon:

t0(t) = yes$(ord(t) eq 1);
tl(t) = yes$(ord(t) eq card(t));
tterm(tt) = yes$(ord(tt) eq card(tt));

* Calibrate the model to the baseline growth path:

y0 = 100;
kd0 = kvs * y0;
l0 = y0 - kd0;
k0 = kd0 / (r + delta);
i0 = (g + delta) * k0;
c0 = y0 - i0;
taxk(t) = 0;
qref(t) = (1+g)**(ord(t)-1);
pref(tt) = (1/(1+r))**(ord(tt)-1);

DISPLAY y0, kd0, l0, k0, i0, c0, g, r, delta;
```

바. GAMS Code: Benchmark Replication of the Model

```

* Define the time horizon of the model using two sets, one which
* includes the post-terminal year and one which covers only the
* endogenous years:

SET tt Time horizon (with the first year of the post-terminal period) /2004*2081/,
t(tt) Time period over the model horizon /2004*2080/;
t0(t) Year 0,
tl(t) Last endogenous year,
tp(tt) First post-terminal year;

$ONTEXT

```

```
$MODEL:RAMSEY

$SECTORS:
Y(t) ! Output
I(t) ! Investment
K(t) ! Capital stock

$COMMODITIES:
P(t) ! Output price
RK(t) ! Return to capital
PK(tt) ! Capital price
PL(t) ! Wage rate

$CONSUMERS:
RA ! Representative agent

$AUXILIARY:
TK ! Post-terminal capital stock

$PROD:Y(t) s:1
O:P(t) Q:Y0
I:PL(t) Q:L0
I:RK(t) Q:KD0 A:RA T:TaxK(t)

$PROD:K(tt)$T(tt)
O:PK(TT+1) Q:(K0*(1-DELTA))
O:RK(tt) Q:KD0
I:PK(tt) Q:K0

$PROD:I(tt)$T(tt)
O:PK(TT+1) Q:I0
I:P(tt) Q:I0

$DEMAND:RA s:1
D:P(t) Q:(qref(t)*C0) P:pref(t)

E:PL(t) Q:(L0*qref(t))
E:PK(T0) Q:(K0*KSTOCK)
E:PK(TP) Q:-1 R:TK

$REPORT:
```

```
V:C(t)  D:P(t)  DEMAND:RA
V:W  W:RA

$CONSTRAINT:TK
SUM(T$TL(T+1),  I(T+1)/I(t)  -  Y(T+1)/Y(t))  =E=  0;

$OFFTEXT
$SYSINCLUDE  mpsgeset  RAMSEY

*  Assign  steady-state  equilibrium  values  for  quantities
*  and  prices:

Y.L(t)  =  qref(t);
I.L(t)  =  qref(t);
K.L(t)  =  qref(t);

P.L(t)  =  pref(t);
RK.L(t)  =  pref(t);
PL.L(t)  =  pref(t);

*  The  steady-state  price  of  capital  is  the  output  price
*  times  one  plus  the  interest  rate:

PK.L(tt)  =  (1+r)  *  pref(tt);
TK.L  =  k0  *  (1+g)**card(t);

RAMSEY.ITERLIM  =  0;
$INCLUDE  RAMSEY.GEN
SOLVE  RAMSEY  USING  MCP;
ABORT$(RAMSEY.OBJVAL  >  0.001)  "Model  does  not  calibrate.";
RAMSEY.ITERLIM  =  10000;
```

3. GAMS-MPSGE modelling

가. Mathematical Formulation

Mathiesen (1985) demonstrated that an Arrow-Debreu general economic equilibrium model could be formulated and efficiently solved as a complementarity problem. Mathiesen's formulation may be posed in terms of three sets of 'central variables':

p	a non-negative n-vector of commodity prices including all final goods, intermediate goods and primary factors of production
y	a non-negative m-vector of activity levels associated with constant returns to scale production sectors in the economy
M	an h-vector of income levels, one for each "household" in the model, including any government entities.

(영이윤 조건)

$$-\Pi_j(p) = C_j(p) - R_j(p) \geq 0 \qquad \forall j$$

$\Pi_j(p)$: unit profit fn

$C_j(p)$: unit cost

$R_j(p)$: unit revenue

$$C_j(p) \equiv \min\left\{\sum_i p_i x_i | f_j(x) = 1\right\}$$

$$R_j(p) \equiv \max\left\{\sum_i p_i y_i | g_j(y) = 1\right\}$$

where f and g (joint output: CET) are the associated production functions characterizing feasible inputs and outputs.

$$f(x) = \phi \prod_i x_i^{\alpha_i} \qquad \sum_i \alpha_i = 1,\ \alpha_i \geq 0$$

$$g(y) = \psi \max_i \frac{y_i}{\beta_i} \qquad \beta_i \geq 0$$

Then the dual functions will be

$$C(p) = \frac{1}{\phi}\prod_i \left(\frac{p_i}{\alpha_i}\right)^{\alpha_i}$$

and

$$R(p) = \sum_i \beta_i p_i$$

(시장청산조건)

at equilibrium prices and activity levels, the supply of any commodity must balance or exceed excess demand by consumers.

$$\sum_j y_j \frac{\partial \Pi_j(p)}{\partial p_i} + \sum_h \omega_{ih} \geq \sum_h d_{ih}(p, M_h)$$

the first term of LHS: the net supply of good i by the CRS production sectors (Shepard's lemma)

the second term of LHS: the aggregated initial endowment of good i by households

RHS: aggregate final demand for good i by households, given market prices p and household income levels M.

Final demand are derived from budget-constraint utility maximization:

$$d_{ih}(p, M_h) = \mathrm{argmax}\left\{U_h(x) | \sum_i p_i x_i = M_h\right\}$$

(소득정의식)

at an equilibrium, the value of each agent's income must equal the value of factor endowments:

$$M_h = \sum_i p_i w_{ih}$$

Walras' law always hold:

$$\sum_i p_i d_{ih} = M_h = \sum_i p_i w_{ih}$$

Aggregating market clearance conditions (using equil prices) and the zero profit conditions (using equi. activity levels), it then follows:

$$\sum_j y_j \Pi_j(p) = 0 \qquad \text{or}$$
$$y_j \Pi_j(p) = 0 \qquad \forall j$$

Furthermore, it follows that:

$$p_i \left(\sum_j y_j \frac{\partial \Pi_j(p)}{\partial p_i} + \sum_h \omega_{ih} - \sum_h d_{ih}(p, M_h) \right) = 0 \qquad \forall i$$

Complementary slackness is a feature of the equilibrium allocation even though it is not imposed as an equilibrium condition, *per se*. This means that in equilibrium, any production activity (which is operated) makes zero profit and any production activity which earns a negative net return is idle. Likewise, any commodity which commands a positive price has a balance between aggregate supply and demand, and any commodity in excess supply has an equilibrium price of zero.

나. Example 1

Table I. The benchmark social accounting matrix.

	Sectors		Consumers		
	X	Y	OWNERS	WORKERS	GOVT
PX	100	-20	-30	-50	
PY	-10	80	-40	-30	
PK	-20	-40	60		
PL	-50	-10		60*	
TRNS	10	20			-30
TK	-20	-10			30

* 60 = labor endowment net leisure demand = 100 - 40

Table II. Mapping from benchmark social accounts into parameters.

	Sectors (S)	Consumers: Households(H)	Government
Goods Markets (G):	A(G,S)-B(G,S)	-C(G,H)	
Factor Markets (F):	-FD(F,S)	E(F,H)-D(F,H)	
Capital taxes:	-T("K",S)		GREV
Transfers:		TRN(H)	-GREV

(1) Gams code: Data specification for the 2-2 model Harberger

```
*        SECTION (i)      DATA SPECIFICATION AND BENCHMARKING

SETS          G           GOODS AND SECTORS /X, Y/,
              F           PRIMARY FACTORS    /K, L/,
              H           HOUSEHOLDS      /OWNER, WORKER/;

ALIAS (S,G);

TABLE SAM(*,*)  SOCIAL ACCOUNTING MATRIX

                X       Y       OWNER   WORKER  GOVT
X               100     -20     -30     -50
Y               -10      80     -40     -30
K               -20     -40      60
L               -50     -10              60
TK              -20     -10                      30
TRN                              10      20     -30
```

```
PARAMETER
        A(S)             BENCHMARK OUTPUT
        B(G,S)           USE MATRIX (GOODS INPUTS BY SECTOR)
        C(G,H)           HOUSEHOLD DEMAND
        FD(F,S)          FACTOR DEMAND BY SECTOR
        E(F,H)           FACTOR ENDOWMENTS
        D(F,H)           FACTOR DEMAND BY HOUSEHOLDS
        T(F,S)           TAX PAYMENT BY FACTOR BY SECTOR
        TRN(H)           TRANSFER REVENUE
        ELAS(S)          ELASTICITY OF SUBSTITUTION IN PRODUCTION
        ESUB(H)          ELASTICITY OF SUBSTITUTION IN DEMAND
        GREV             BENCHMARK GOVERNMENT REVENUE
        TF(F,S)          FACTOR TAX RATE
        PF(F,S)          BENCHMARK FACTOR PRICES GROSS OF TAX
        THETA(G)         WEIGHTS IN NUMERAIRE PRICE INDEX
        WBAR(H)          BENCHMARK WELFARE INDEX;

*       EXTRACT DATA FROM THE SOCIAL ACCOUNTING MATRIX:

A(S)    = SAM(S,S);
B(G,S) = MAX(0, -SAM(G,S));
C(G,H) = -SAM(G,H);
FD(F,S) = -SAM(F,S);
E(F,H) = SAM(F,H);
D(F,H) = 0;
TRN(H) = SAM("TRN",H);
T("K",S) = -SAM("TK",S);

*       INSTALL "FREE" ELASTICITY PARAMETERS:

E("L","WORKER") = 100;
D("L","WORKER") =  40;
ELAS(S) = 1;
ESUB(H) = 0.5;

*       INSTALL FUNCTIONS OF BENCHMARK VALUES:

GREV      = SUM(H, TRN(H));
TF(F,S) = T(F,S) / FD(F,S);
PF(F,S) = 1 + TF(F,S);
THETA(G) = SUM(H, C(G,H));
THETA(G) = THETA(G) / SUM(S, THETA(S));
WBAR(H) = SUM(G, C(G,H)) + SUM(F, D(F,H));
```

(2) Gams code: MPSGE model specification and benchmark replication

```
*       SECTION (ii)    MPSGE MODEL DECLARATION

$ONTEXT

$MODEL:HARBERGER

$SECTORS:
      AL(S)

$COMMODITIES:
      P(G)  W(F)  PT

$CONSUMERS:
      RA(H) GOVT

$PROD:AL(S)  s:0  a:ELAS(S)
      O:P(S)       Q:A(S)
      I:P(G)       Q:B(G,S)
      I:W(F)       Q:FD(F,S)   P:PF(F,S)   A:GOVT   T:TF(F,S)  a:

$DEMAND:RA(H)  s:1  a:ESUB(H)
      D:P(G)       Q:C(G,H)   a:
      D:W(F)       Q:D(F,H)
      E:W(F)       Q:E(F,H)
      E:PT         Q:TRN(H)

$DEMAND:GOVT
      D:PT         Q:GREV

$REPORT:
        V:CD(G,H)        D:P(G)           DEMAND:RA(H)
        V:DF(F,H)        D:W(F)           DEMAND:RA(H)
        V:EMPLOY(S)      I:W("L")         PROD:AL(S)
        V:WLF(H)         W:RA(H)

$OFFTEXT
```

다. Example 2

$PROD:Y s:0.75 va:1
O:PY Q:120
I:PX Q: 20
I:PL Q: 60 va:

I:PK Q: 40 va:



$PROD:X s:esub id:0 e:sigma kl(e):1 en(e):1
O:PX Q:x0
I:P(g) Q:yx0(g) id:$(not en(g)) en:$en(g)
I:PL Q:lx0 P:pl0 kl:

I:PK   Q:kx0

P:pk0 kl:



4. Global SAM

		1 Imported commodi	2 Domesti c	3 Activitie s	4 Factors	5 Regional househo	6 Private househo	7 Trade taxes	8 Import sales	9 Domesti c sales	10 Factor taxes	11 Producti on taxes	12 Direct taxes	13 Governm ent	14 Capital	15 Import margins	16 Export margins	17 Rest of world	18 Totals
1	Imported commoditi			c*a			c*1							c*1	c*1				c*1
2	Domestic commoditi			c*a			c*1							c*1	c*1		c*m	c*k	c*1
3	Activities		a*c																c*1
4	Factors			f*a															f*1
5	Regional household				1*f			1*c	1*c	1*c	1*f	1*1	1*1						1*1
6	Private household					1*1													1*1
7	Trade taxes	2k*c	2k*c																2k*1
8	Import sales			c*a			c*1							c*1	c*1				c*1
9	Domestic sales			c*a			c*1							c*1	c*1				c*1
10	Factor taxes			f*a															f*1
11	Production taxes			1*a															1*1
12	Direct taxes				1*f														1*1
13	Governme nt					1*1													1*1
14	Capital				1*f	1*1											1*m	1*k	1*1
15	Import margins	m*k*c																	m*k*1
16	Export margins															m*m*k			m*1
17	Rest of world	k*c																	k*1
18	Totals	1*c	1*c	1*a	1*f	1*1	1*1	1*2k	1*c	1*c	1*f	1*1	1*1	1*1	1*1	1*m*k	1*m	1*k	